```
Show[Graphics[{LightGray, Rectangle[{-10, -10}, {10, 10}], Red, Disk[{0, 0}, 10],
   Darker@Red, Rectangle[{-10, -10} / Sqrt[2], {10, 10} / Sqrt[2]],
   Black, Point /@ Tuples[Range[-10, 10], 2]}]]
```

Out[238]=



In[230]:=
```
(*whether a rectangle contains a lattice point*)
containslattice[{{x0_, y0_}, {x1_, y1_}}] :=
  (Ceiling[x0] ≤ Floor[x1]) && (Ceiling[y0] ≤ Floor[y1]);
(*lattice points in a rectangle*)
lattice[{{x0_, y0_}, {x1_, y1_}}] :=
  Tuples[{Range[x0 // Ceiling, x1 // Floor], Range[y0 // Ceiling, y1 // Floor]}];
(*cases for expanding beyond the inscribed square*)
expandup[{a_, b_}, r_, points_] :=
  Module[{limits = Floor[b + Sqrt[r^2 - (# - a)^2]] & /@ points},
   Catenate[Tuples[{{points[[#]]}, Range[Floor[b + r / Sqrt[2]] + 1, limits[[#]]]}] & /@
     Range[Length[points]]]
  ];
expanddown[{a_, b_}, r_, points_] :=
  Module[{limits = Ceiling[b - Sqrt[r^2 - (# - a)^2]] & /@ points},
   Catenate[Tuples[{{points[[#]]},
        Range[limits[[#]], Ceiling[b - r / Sqrt[2]] - 1]}] & /@ Range[Length[points]]]
  ];
expandright[{a_, b_}, r_, points_] :=
  Module[{limits = Floor[a + Sqrt[r^2 - (# - b)^2]] & /@ points},
   Catenate[Tuples[{Range[Floor[a + r / Sqrt[2]] + 1, limits[[#]]], {points[[#]]}}] & /@
     Range[Length[points]]]
```

```
  ];
expandleft[{a_, b_}, r_, points_] :=
  Module[{limits = Ceiling[a - Sqrt[r^2 - (# - b)^2]] & /@ points},
   Catenate[Tuples[{Range[limits[[#]], Ceiling[a - r / Sqrt[2]] - 1],
        {points[[#]]}}] & /@ Range[Length[points]]]
  ];
circle[center_, r_] := Module[{inscribedsquare, vargs, hargs, points,
    circleequation = (#[[1]] - center[[1]])^2 + (#[[2]] - center[[2]])^2 ≤ r^2 &},
   inscribedsquare = {(center - r / Sqrt[2]) // Ceiling,
     (center + r / Sqrt[2]) // Floor};
   (*handles edge cases for small circles*)
   If[! containslattice[inscribedsquare],
    Return[Select[lattice[{center - r, center + r}], circleequation]]];
   vargs = {center, r, Range @@ First /@ inscribedsquare};
   hargs = {center, r, Range @@ Last /@ inscribedsquare};
   points = Join[lattice[inscribedsquare], expandup @@ vargs,
     expanddown @@ vargs, expandright @@ hargs, expandleft @@ hargs, 1];
   Print[Graphics[{LightGray, Rectangle[center - r, center + r], Red,
      EdgeForm[Directive[Thick, Black]], Disk[center, r], Darker@Red, EdgeForm[],
      Rectangle @@ inscribedsquare, Black, Point /@ lattice[{center - r, center + r}],
      White, Large // PointSize, Point /@ points}]];
   Print["Number of points: ", Length[points]];
   Print["Are all points in the circle? ", AllTrue[points, circleequation]];
   Print["Are they unique? ", DuplicateFreeQ[points]];
   Print["Are they integers? ", AllTrue[points, IntegerPart[#] == # &]];
   Print["Does this set contain all of the points in a circle? ", Sort@points ==
     Sort@Select[lattice[{center - r, center + r}], Element[#, Disk[center, r]] &]];
   points
  ];
circle[{0, 0}, 10];
```



```
Number of points: 317
Are all points in the circle? True
Are they unique? True
Are they integers? True
Does this set contain all of the points in a circle? True
```